15 moksleivių informatikos olimpiada

3 etapo 1 dalis

Sprendimai

2862. Akcijos.

Pirmiausia atkreipkite dėmesį, kad akcijas galima pirkti ir parduoti keletą kartų. Pavyzdžiui, pirmą dieną pirkti akcijas po vieną litą, antrą dieną jas parduoti po penkis litus, o trečią vėl pirkti po du litus ir t.t.

Akcijų kursą per visą periodą, galima pavaizduoti kaip kreivę, kai akcijos tai kyla, tai krenta:

Taškai, kuriuose akcijų kursas nustoja augti ir ima mažėti vadinami lokaliaisiais maksimumais, o taškai, kuriuose akcijų kursas nustoja mažėti ir ima augti – lokaliaisiais minimumais. Akcijos perkamos lokaliuosiuose minimumuose, o parduodamo lokaliuosiuose maksimumuose. 

Paveikslėlyje pateiktu atveju akcijos bus perkamos už 10 litų, parduodamos už 40 litų, toliau vėl perkamos už 15 litų, parduodamos už 40 litų, perkamos už 5, parduodamos už 12 litų, perkamos už 1, parduodamos už 20 litų. 

Gali kilti klausimas ar tikrai verta pirkti (parduoti) kiekviename lokaliame minimume (maksimume). 

Tarkime, kad nupirkę akcijas lokalaus minimumo taške nenorime jų parduoti artimiausiame lokalaus maksimumo taške, o norime parduoti tolesniame maksimumo taške. (Pvz. akcijas pirkome už 1 litą, jų nenorime parduoti už 10 litų pirmame maksimumo taške, o norime parduoti už 100 litų kitame lokalaus maksimumo taške).

Sakykime, akcijų kainos atitinkamai lygios: 

· xi (i-as lokalus minimumas), ;

· yi (i-as lokalusis maksimumas), ;

Sakykime, esame nupirkę A akcijų. Jas pardavę artimiausiame maksimume ir vėl pirkę artimiausiame minimume turėsime: A(y1 div x2 akcijų. Pinigų likučio, kuris liko pirkus A akcijų, nepanaudojome. Kadangi y1 > x2, tai ir A(y1 div x2 ( A. O jei akcijų nenorėsime parduoti artimiausiame lokaliame maksimume, tuomet turėsime tik A akcijų. Tad akivaizdu, kad apsimoka pirkti/parduoti kiekviename lokaliame minimume/maksimume.

Na o rašant algoritmą reikia nepamiršti pinigų likučio, likusio perkant akciją. Likutį kiekvieną kartą pardavus akcijas reikia pridėti prie gautos už akcijas pinigų sumos. 

Uždavinį sprendžianti procedūra bus paprastesnė, jeigu atskiromis procedūromis aprašysime akcijų pirkimą ir pardavimą:

procedure pirkti (kaina: integer; var A, P: integer);

{ už turimus pinigus P perkamos akcijos; nupirktų akcijų skaičius lygus A }

begin

  A := P div kaina;  P := P mod kaina;

end; { pirkti }
procedure parduoti (kaina: integer; var A, P: integer);
{ parduodama A akcijų;  }

begin

  P := P + A*kaina;

  A := 0;

end; { parduoti }
Žemiau pateikta uždavinį sprendžianti procedūra. Pradinių duomenų (akcijų kainų per visą periodą) nebūtina įsiminti. Užtenka žinoti tris paskutiniąsias skirtingas kainas.

procedure prekiauti (var f: text; var P: integer; N: integer);

  var A, kaina1, kaina2, kaina3, d : integer;

begin

  readln (F, kaina1, kaina2);

  A := 0; { nupirktų akcijų skaičius }
  if kaina1 < kaina2

     then pirkti (kaina1, A, P);

  for d := 3 to N do
    begin

      readln (f, kaina3);

      if (kaina1 < kaina2) and (kaina2 > kaina3)

         then parduoti (kaina2, A, P)

      else if (kaina1 > kaina2) and (kaina2 < kaina3)

         then pirkti (kaina2, A, P)

      else if (A > 0) and (d = N) { paskutinioji periodo diena }
         then parduoti (kaina3, A, P);

      if kaina2 <> kaina3

         then kaina1 := kaina2; 

      kaina2 := kaina3;

    end; 

  if A > 0 { paskutinioji periodo diena }
     then parduoti (kaina2, A, P)  

end; { prekiauti }
Žemiau pateikta uždavinį sprendžianti procedūra. Pradinių duomenų (akcijų kainų per visą periodą) nebūtina įsiminti. Užtenka žinoti tris paskutiniąsias skirtingas kainas.

287. Rikis.

Šis uždavinys sprendžiamas Bangos metodu. Iš pradinio langelio (jame stovi Rikis) visomis įmanomomis kryptimis paeinamas vienas žingsnis. Visuose šiuose langeliuose įrašomas vienetukas (langelis pasiekiamas per vieną žingsnį). Toliau peržiūrima lenta ir vėl visomis kryptimis bandoma eiti iš visų langelių, kuriuose įrašytas vienetukas. Naujai pasiektuose langeliuose įrašomas dvejetukas (langeliai pasiekiami per du žingsnius). 

Veiksmai kartojami kol pasiekiama valdovė arba nebegalima pasiekti naujų langelių. 

Šį metodą galima realizuoti dviem būdais. Pirmasis būdas: kiekvieną kartą iš naujo peržiūrėti visą lentą ir pirmą kartą analizuoti ėjimus iš visų langelių, kuriuose įrašyti vienetai, antrą kartą peržiūrint lentą analizuoti ėjimus iš langelių, kuriuose įrašyti dvejetai ir t.t.

Antrasis būdas – eilėje (pvz. masyve) saugoti langelių, iš kurių dar reikia nagrinėti ėjimus, koordinates. T.y. pradžioje eilėje bus saugomos langelio, kuriame stovi rikis, koordinatės. Išnagrinėjus visus ėjimus iš šio langelio, jis pašalinamas iš eilės, o visi naujai pasiekti langeliai įtraukiami į eilę. Taip peržiūrima eilė, kol pasiekiamas valdovės langelis arba eilė ištuštėja.

Paanalizuokime abiejų algoritmų efektyvumą. Imkime pirmąjį algoritmą (kai kas kartą peržiūrima lenta). Jei lentos dydis N, vienas lentos peržiūrėjimas blogiausiu atveju užims iki O(N2/2) operacijų (peržiūrimi tik juodi arba tik balti langeliai). Jei laikytume, kad blogiausiu teks peržiūrėti kelią iš visų lentos langelių, tuomet lentą peržiūrėti tektų N2/2 kartų. Tad blogiausiu atveju operacijų skaičius yra O(N4/4) eilės. Kai lenta nedidelė (N(8, jaunesniųjų uždavinyje), toks algoritmas bus pakankamai efektyvus. 

Kai N didesnis (N(100 vyresniųjų grupės uždavinyje), tuomet algoritmas gali veikti nepakankamai greitai. 

Antrasis algoritmas (kai sudaroma eilė) yra O(N2/2) eilės: blogiausiu atveju į eilę įtraukiami visi lentos langeliai. N artėjant prie šimto algoritmas bus pakankamai efektyvus. Tačiau šis algoritmas naudoja daugiau atminties. Blogiausiu atveju eilėje gali būti ne daugiau pusės langelių (rikis vaikšto tik juodais arba tik baltais langeliais), t.y. N2/2 langelių. Kai N(100, tai būtų eilė, kurios ilgis neviršija 5000 elementų. Atminties ribojimai (iki 16 MB) puikiausiai leidžia sudaryti tokio ilgio masyvus. 

Tad esant nedideliam N (N(8, jaun. uždavinyje), visiškai nesvarbu kuris algoritmas bus pasirinktas. O esant didesniam N (N(100), efektyvesnis antrasis algoritmas.

{ uždavinys sprendžiamas bangos metodu; banga įsimenama eilėje }

const  MAX = 100;  { maksimalus lentos dydis }
       ME = MAX*MAX div 2 + 1;

type lenta = array [1..MAX, 1..MAX] of integer;

     lang = record
              i, j: integer;

            end;

     eilė = array [1..ME] of lang;

{ žemiau aprašyti globalieji kintamieji }

var l: lenta;

    N: integer;

procedure į_eilę(lng: lang; var pb: integer; var E: eilė);

begin

  E[pb] := lng;

  if pb = ME

     then pb := 1

     else pb := pb + 1;

end; { į_eilę }
procedure iš_eilės (var pr: integer);

begin

  if pr = ME

     then pr := 1

     else pr := pr + 1;

end; { iš_eilės }
procedure eiti (lng: lang; viršun, kairėn: boolean;

                var pr, pb: integer; var E: eilė); { kuria kryptimi }
  { paeina nurodyta kryptimi }

  var i, j, di, dj, žings: integer;

      baigti: boolean;

      lan: lang;

begin

  if viršun then di := -1

            else di := 1;

  if kairėn then dj := -1

            else dj := 1;

  žings := L[lng.i, lng.j]; baigti := false;

  i := lng.i; j := lng.j;

  while (i+di >= 1) and (i+di <= N) and
        (j+dj >= 1) and (j+dj <= N) and not baigti do
    begin

      i := i + di; j := j + dj;

      case L[i, j] of
       -1: begin
             L[i, j] := žings + 1; { galima eiti }
             lan.i := i; lan.j := j;

             į_eilę (lan, pb, E);

           end;

       -3: baigti := true; { negalima eiti }
      end;

    end;

end; { eiti }
function ieškoti (R, V: lang): integer;

  { pagrindinė funkcija, randanti ėjimų skaičių }

  var pr, pb: integer;

      E: eilė;

      lng: lang;

begin

  pr := 1; pb := 1;

  į_eilę (R, pb, E);

  while (pr <> pb) and (L[V.i, V.j] = -1) do
    begin

      lng := E[pr];

      eiti (lng, true, true, pr, pb, E);

      eiti (lng, true, false, pr, pb, E);

      eiti (lng, false, true, pr, pb, E);

      eiti (lng, false, false, pr, pb, E);

      iš_eilės (pr);

    end;

  if L[V.i, V.j] = -1 then ieškoti := 0

     else ieškoti := L[V.i, V.j];

end; { ieškoti }
288. HTML sintaksės tikrinimas.

Pastebėsime, kad savybės ir gaires gaubiantys skliaustai uždavinio spendimui jokios įtakos neturi. Tad paprasčiausia perskaičius kiekvieną gairę nuimti gaubiančius skliaustus ir savybes. Tai atlieka ši procedūra:

procedure sutrumpinti (gairė: string; var ga: string);

  { pašalinami skliaustai ir savybės }

  var tarpas: integer;

begin

  tarpas := pos(' ', gairė);

  if tarpas > 0

     then ga := copy (gairė, 2, tarpas-2)

     else ga := copy (gairė, 2, length(gairė)-2);

end; { sutrumpinti }
Pereiname prie uždavinio sprendimo. Esmė – sumodeliuoti dėklą. Dėklas tai LIFO (Last in First out) tipo duomenų struktūra. Tai tokia eilė, kur paskutinysis atsistojęs į eilę, pirmasis iš jos išeina.

Kiekviena naujai perskaityta atidarančioji gairė dedama į dėklą. Jei gairė yra uždarančioji, tuomet paskutinioji į dėklą įdėta gairė turi būti ją atitinkanti atidarančioji gairė. Jei taip ir yra, tuomet pašaliname paskutiniąją dėklo gairę ir gairės nagrinėjamos toliau. Jei taip nėra, radome pirmąją nekorektiškoje vietoje esančią gairę. 

Jei peržiūrėjome visas gaires, tačiau dėklas nėra tuščias, tai reiškia, kad yra atidarytų, bet neuždarytų blokų. Tokiu atveju rezultatas turi būti NEGERAI.

Panagrinėkime konkretų pavyzdį, kaip dirbama su dėklu.

Prieš pradedant darbą.

	Neišnagrinėtos  gairės
	Dėklas

	<body1 bgcolor=blue>

<p font=verdana size=+3>

</p>

</body1>

</body1>
</html>
	

	
	

	
	

	
	

	
	

	
	html


Išnagrinėta 1 gairė

	Neišnagrinėtos  gairės
	Dėklas

	<p font=verdana size=+3>

</p>

</body1>

</body1>
</html>
	

	
	

	
	

	
	

	
	body1

	
	html


Išnagrinėtos 2 gairės

Išnagrinėtos 4 gairės

Išnagrinėtos 5 gairės

Paėmę 6-ąją gairę (</body1>) matome, kad ji yra bloką body1 uždarančioji gairė. Paskutinė dėkle esanti gairė nėra šį bloką atidaranti gairė. Tad radome nekorektišką eilutė – gairę, uždarančią neatidarytą bloką. 

procedure HTML_analizė (var f: text; N: integer; var rz: string);

  var pab, i: integer;

      d: dėklas;

      gerai: boolean;

      gairė, ga: string;

begin

  pab := 0; d[0] := '';

  i := 1; gerai := true;

  while (i <= N) and gerai do
    begin

      i := i + 1;

      readln (f, gairė);

      sutrumpinti (gairė, ga);

      if ga[1] = '/' { jei uždaranti gairė }
         then if '/'+ d[pab] = ga { jei blokas uždaromas }
                 then pab := pab - 1 { išimama iš dėklo }
                 else begin

                        gerai := false;

                        ga := gairė;

                      end

         else begin { jei atidaranti gairė }
                pab := pab + 1;

                d[pab] := ga;

              end;

    end;

  if gerai

     then if pab <> 0 { jei dėklas netuščias }
             then rz := 'NEGERAI'

             else rz := 'GERAI'

end; { tikrinti }
Šioje procedūroje dėklas realizuotas masyvu. Kita galimybė – dėklą realizuoti dinaminėmis duomenų struktūromis.

